METHOD IF_UJ_CUSTOM_LOGIC~EXECUTE.

* Create a structure same as D_SALES application

  DATA: BEGIN OF SALES_STRUC,

           CATEGORY(20)   TYPE C,

           PRODUCT(20)  TYPE C,

           KFTYPE(20)     TYPE C,

           CUSTOMER(20)  TYPE C,

           ENTITY(20)   TYPE C,

           TIME(20)       TYPE C,

           SIGNEDDATA     TYPE P DECIMALS 5,

        END OF SALES_STRUC.

* Create a structure same as D_PROMOTION application

  DATA: BEGIN OF PROMOTION_STRUC,

           CATEGORY(20)   TYPE C,

           PRDGRP(20)  TYPE C,

           KFTYPE(20)     TYPE C,

           ENTITY(20)   TYPE C,

           TIME(20)       TYPE C,

           SIGNEDDATA     TYPE P DECIMALS 5,

        END OF PROMOTION_STRUC.

  DATA: BEGIN OF PRODUCT_STRUC,

         PRODUCT(20)  TYPE C,

         PRDGRP(20) TYPE C,

      END OF PRODUCT_STRUC.

*** BPC variables

  DATA :

          L_APPSET_ID TYPE UJ_APPL_ID,

          LO_MODEL1 TYPE REF TO IF_UJ_MODEL,

          LO_MODEL2 TYPE REF TO IF_UJ_MODEL,

          L_APPL_ID1 TYPE UJ_APPSET_ID,

          L_APPL_ID2 TYPE UJ_APPSET_ID,

          LO_DIM_PRODUCT       TYPE REF TO CL_UJA_DIM,

          LT_PROMOTION_RG_SEL TYPE UJ0_T_SEL,

          LI_RG_SEL LIKE LINE OF LT_PROMOTION_RG_SEL,

          LS_PROMOTION_ARRAY TYPE REF TO DATA,

          L_SECURITYCHECK TYPE UJ_FLG,

          L_FIRST_CALL  TYPE RS_BOOL,

          LT_PROMOTION_DIMS TYPE UJQ_T_DIM,

          LI_PROMOTION_DIM LIKE LINE OF LT_PROMOTION_DIMS,

          L_USER TYPE UJ0_S_USER,

          LI_ATTR               TYPE LINE OF UJA_T_ATTR_NAME,

          LT_ATTR_PRODUCT      TYPE UJA_T_ATTR_NAME,

          LT_SEL_PRODUCT_MBR   TYPE UJA_T_DIM_MEMBER,

          LI_SEL_MBR            TYPE LINE OF UJA_T_DIM_MEMBER,

          STAT_ID   TYPE UJ_STAT_SESSION.

  DATA:

     L_LOG TYPE STRING,

     LR_RESULT_REC TYPE REF TO DATA,

     LR_NEW_REC TYPE REF TO DATA,

     LT_FINAL      TYPE REF TO DATA,

     LT_SALES      LIKE STANDARD TABLE OF SALES_STRUC,

     LI_SALES      LIKE LINE OF LT_SALES,

     LT_PROMOTION  LIKE STANDARD TABLE OF PROMOTION_STRUC,

     LI_PROMOTION  LIKE LINE OF LT_PROMOTION,

     L_DISCNTPCT   TYPE P DECIMALS 5,

     L_ACTPRICE    TYPE P DECIMALS 5,

     L_calc        TYPE P DECIMALS 5,

     LS_ARRAY      TYPE REF TO DATA,

     LT_PRODUCT    LIKE STANDARD TABLE OF PRODUCT_STRUC,

     LI_PRODUCT    LIKE LINE OF LT_PRODUCT.

  FIELD-SYMBOLS: <LS_REC> TYPE ANY,

    <LS_RESULT_REC> TYPE ANY,

    <LS_NEW_REC> TYPE ANY,

    <LS_TIME>       TYPE ANY,

    <LS_PRODUCT>  TYPE ANY,

    <LS_CUSTOMER>   TYPE ANY,

    <LS_KFTYPE>    TYPE ANY,

    <LS_PRDGRP>  TYPE ANY,

    <LS_ENTITY>   TYPE ANY,

    <LS_CATEGORY>   TYPE ANY,

    <LS_SIGNEDDATA> TYPE ANY,

    <LS_TIME_NEW>       TYPE ANY,

    <LS_PRODUCT_NEW>  TYPE ANY,

    <LS_CUSTOMER_NEW>   TYPE ANY,

    <LS_KFTYPE_NEW>    TYPE ANY,

    <LS_ENTITY_NEW>   TYPE ANY,

    <LS_CATEGORY_NEW>   TYPE ANY,

    <LS_SIGNEDDATA_NEW> TYPE ANY,

    <LS_MEMBER>     TYPE ANY,

    <LI_MEMBER>       TYPE ANY,

    <LT_FINAL>      TYPE STANDARD TABLE,

    <LT_PROMOTION>   TYPE STANDARD TABLE,

    <LS_PROMOTION>   TYPE ANY,

    <LI_PROMOTION>   TYPE ANY,

    <LT_DIM_PRD_MBR>  TYPE ANY TABLE.

  L_APPSET_ID = 'DEMO3'. 

  L_APPL_ID1  = 'D_SALES'.

  L_APPL_ID2  = 'D_PROMOTION'.

  L_SECURITYCHECK = ''.

  L_FIRST_CALL = ''.

  L_USER = 'BWEXPERT'.

** List of attributes to be read from D_PRODUCT Dimension. We need to read only one 

** attribute you can add more lines to the LT_ATTR_PRODCT table if you need to read more 

** attributes/properties

  LI_ATTR = 'D_PRDGRP'.

  APPEND LI_ATTR TO LT_ATTR_PRODUCT.

  TRY.

      CREATE OBJECT LO_DIM_PRODUCT

        EXPORTING

          I_APPSET_ID = L_APPSET_ID

          I_DIMENSION = 'D_PRODUCT'.

    CATCH CX_UJA_ADMIN_ERROR.

      MESSAGE I001(00) WITH TEXT-011.

      RETURN.

  ENDTRY.

  TRY.

      CALL METHOD LO_DIM_PRODUCT->CREATE_DATA_REF

        EXPORTING

          I_DATA_TYPE = 'T'

        IMPORTING

          ER_DATA     = LS_ARRAY.

      ASSIGN LS_ARRAY->* TO <LT_DIM_PRD_MBR>.

    CATCH CX_UJA_ADMIN_ERROR.

      MESSAGE I001(00) WITH TEXT-011.

      RETURN.

  ENDTRY.

  CLEAR LS_ARRAY.

**Instead of reading the product group for each record

** Read all material and assign to an internal table for performance

  TRY.

      CALL METHOD LO_DIM_PRODUCT->READ_MBR_DATA

        EXPORTING

          IT_ATTR_LIST = LT_ATTR_PRODUCT

*            it_sel_mbr   = lt_sel_material_mbr

        IMPORTING

          ER_DATA      = LS_ARRAY.

    CATCH CX_UJA_ADMIN_ERROR.

      MESSAGE I001(00) WITH TEXT-011.

      RETURN.

  ENDTRY.

  ASSIGN LS_ARRAY->* TO <LT_DIM_PRD_MBR>.

  UNASSIGN: <LI_MEMBER>,<LS_PRODUCT>,<LS_PRDGRP>.

  LOOP AT <LT_DIM_PRD_MBR> ASSIGNING <LI_MEMBER>.

    ASSIGN COMPONENT 'ID' OF STRUCTURE <LI_MEMBER> TO <LS_PRODUCT>.

    ASSIGN COMPONENT 'D_PRDGRP' OF STRUCTURE <LI_MEMBER> TO <LS_PRDGRP>.

    LI_PRODUCT-PRODUCT = <LS_PRODUCT>.

    LI_PRODUCT-PRDGRP = <LS_PRDGRP>.

    APPEND LI_PRODUCT TO LT_PRODUCT.

  ENDLOOP.

  CREATE DATA LT_FINAL LIKE CT_DATA.

  ASSIGN LT_FINAL->* TO <LT_FINAL>.

  CREATE DATA LR_RESULT_REC LIKE LINE OF CT_DATA.

  ASSIGN LR_RESULT_REC->* TO <LS_RESULT_REC>.

  CREATE DATA LR_NEW_REC LIKE LINE OF CT_DATA.

  ASSIGN LR_NEW_REC->* TO <LS_NEW_REC>.

  LI_PROMOTION_DIM = 'D_CATEGORY'.

  APPEND LI_PROMOTION_DIM TO LT_PROMOTION_DIMS.

  LI_PROMOTION_DIM = 'D_ENTITY'.

  APPEND LI_PROMOTION_DIM TO LT_PROMOTION_DIMS.

  LI_PROMOTION_DIM = 'D_KFTYPE'.

  APPEND LI_PROMOTION_DIM TO LT_PROMOTION_DIMS.

  LI_PROMOTION_DIM = 'D_PRDGRP'.

  APPEND LI_PROMOTION_DIM TO LT_PROMOTION_DIMS.

  LI_PROMOTION_DIM = 'D_TIME'.

  APPEND LI_PROMOTION_DIM TO LT_PROMOTION_DIMS.

*** Loop through the incoming records from D_SALES cube/application. These record are based on your

*** Scope data you defined in your Script Logic using "XDIM_MEMBERSET keyword

  LOOP AT CT_DATA ASSIGNING <LS_RESULT_REC>.

    ASSIGN COMPONENT 'D_TIME' OF STRUCTURE <LS_RESULT_REC> TO <LS_TIME>.

    ASSIGN COMPONENT 'D_CATEGORY' OF STRUCTURE <LS_RESULT_REC> TO <LS_CATEGORY>.

    ASSIGN COMPONENT 'D_PRODUCT' OF STRUCTURE <LS_RESULT_REC> TO <LS_PRODUCT>.

    ASSIGN COMPONENT 'D_KFTYPE' OF STRUCTURE <LS_RESULT_REC> TO <LS_KFTYPE>.

    ASSIGN COMPONENT 'D_ENTITY' OF STRUCTURE <LS_RESULT_REC> TO <LS_ENTITY>.

    ASSIGN COMPONENT 'D_CUSTOMER' OF STRUCTURE <LS_RESULT_REC> TO <LS_CUSTOMER>.

    ASSIGN COMPONENT 'SIGNEDDATA' OF STRUCTURE <LS_RESULT_REC> TO <LS_SIGNEDDATA>.

** Assign the each record to a new record structure, we'll use this data

** to create new entries as all the fields will be the same except the KFTYPE and SigngedData/Measure 
    ASSIGN COMPONENT 'D_TIME' OF STRUCTURE <LS_NEW_REC> TO <LS_TIME_NEW>.

    ASSIGN COMPONENT 'D_CATEGORY' OF STRUCTURE <LS_NEW_REC> TO <LS_CATEGORY_NEW>.

    ASSIGN COMPONENT 'D_PRODUCT' OF STRUCTURE <LS_NEW_REC> TO <LS_PRODUCT_NEW>.

    ASSIGN COMPONENT 'D_KFTYPE' OF STRUCTURE <LS_NEW_REC> TO <LS_KFTYPE_NEW>.

    ASSIGN COMPONENT 'D_ENTITY' OF STRUCTURE <LS_NEW_REC> TO <LS_ENTITY_NEW>.

    ASSIGN COMPONENT 'D_CUSTOMER' OF STRUCTURE <LS_NEW_REC> TO <LS_CUSTOMER_NEW>.

    ASSIGN COMPONENT 'SIGNEDDATA' OF STRUCTURE <LS_NEW_REC> TO <LS_SIGNEDDATA_NEW>.

    <LS_TIME_NEW> = <LS_TIME>.

    <LS_CATEGORY_NEW> = <LS_CATEGORY>.

    <LS_PRODUCT_NEW> = <LS_PRODUCT>.

    <LS_KFTYPE_NEW> = <LS_KFTYPE>.

    <LS_ENTITY_NEW> = <LS_ENTITY>.

    <LS_CUSTOMER_NEW> = <LS_CUSTOMER>.

    <LS_SIGNEDDATA_NEW> = <LS_SIGNEDDATA>.

** For each record find the Product's group by reading the internal table

** and use that for in the selection to read the D_PROMOTION cube/Application

    READ TABLE LT_PRODUCT INTO LI_PRODUCT WITH KEY PRODUCT = <LS_PRODUCT>.

**** Promotion cube selection

    CLEAR:LT_PROMOTION_RG_SEL.

    LI_RG_SEL-SIGN = 'I'.

    LI_RG_SEL-OPTION = 'EQ'.

    LI_RG_SEL-DIMENSION = 'D_CATEGORY'.

    LI_RG_SEL-LOW = <LS_CATEGORY>.

    APPEND LI_RG_SEL TO LT_PROMOTION_RG_SEL.

    LI_RG_SEL-DIMENSION = 'D_ENTITY'.

    LI_RG_SEL-LOW = <LS_ENTITY>.

    APPEND LI_RG_SEL TO LT_PROMOTION_RG_SEL.

    LI_RG_SEL-DIMENSION = 'D_KFTYPE'.

    LI_RG_SEL-LOW = 'DISCOUNT'.

    APPEND LI_RG_SEL TO LT_PROMOTION_RG_SEL.

    LI_RG_SEL-DIMENSION = 'D_PRDGRP'.

    LI_RG_SEL-LOW = LI_PRODUCT-PRDGRP.

    APPEND LI_RG_SEL TO LT_PROMOTION_RG_SEL.

    LI_RG_SEL-DIMENSION = 'D_TIME'.

    LI_RG_SEL-LOW =  <LS_TIME>.

    APPEND LI_RG_SEL TO LT_PROMOTION_RG_SEL.

    UNASSIGN:<LT_PROMOTION>.

    TRY.

        LO_MODEL1 = CL_UJ_MODEL=>GET_MODEL( L_APPSET_ID ).

        CALL METHOD LO_MODEL1->CREATE_TX_DATA_REF

          EXPORTING

            I_APPL_NAME = L_APPL_ID2

            I_TYPE      = 'T'

          IMPORTING

            ER_DATA     = LS_PROMOTION_ARRAY.

      CATCH CX_UJ_STATIC_CHECK.

    ENDTRY.

    ASSIGN LS_PROMOTION_ARRAY->* TO <LT_PROMOTION>.

*** Execute the Function module to read data from the D_PROMOTION cube for the selection

    TRY.

        CALL FUNCTION 'UJQ_RUN_RSDRI_QUERY'

          EXPORTING

            I_APPSET_ID       = L_APPSET_ID

            I_APPL_ID         = L_APPL_ID2

            IT_DIM_NAME       = LT_PROMOTION_DIMS

            IT_SEL            = LT_PROMOTION_RG_SEL

            IF_CHECK_SECURITY = ABAP_FALSE

          IMPORTING

            E_STATS_GUID      = STAT_ID

            ET_DATA           = <LT_PROMOTION>.

      CATCH CX_UJQ_EXCEPTION.

        MESSAGE I001(00) WITH TEXT-011.

        RETURN.

    ENDTRY.

** Loop through the promotion app records to get the discount

    UNASSIGN: <LS_PROMOTION>.

    CLEAR: L_DISCNTPCT,L_ACTPRICE.

    LOOP AT <LT_PROMOTION> ASSIGNING <LS_PROMOTION>.

      ASSIGN COMPONENT 'SIGNEDDATA' OF STRUCTURE <LS_PROMOTION> TO <LI_PROMOTION>.

      L_DISCNTPCT = <LI_PROMOTION>.

      L_ACTPRICE = <LS_SIGNEDDATA>.

** Add two new records to the D_SALES cube

** 1 - DISCOUNT KFTYPE read from the D_PROMTION cube

      <LS_TIME_NEW> = <LS_TIME>.

      <LS_CATEGORY_NEW> = <LS_CATEGORY>.

      <LS_PRODUCT_NEW> = <LS_PRODUCT>.

      <LS_ENTITY_NEW> = <LS_ENTITY>.

      <LS_CUSTOMER_NEW> = <LS_CUSTOMER>.

      <LS_SIGNEDDATA_NEW> = <LI_PROMOTION>.

      <LS_KFTYPE_NEW> = 'DISCOUNT'.

       APPEND <LS_NEW_REC> TO <LT_FINAL>.

** 2 - SALEPRICE KFTYPE calculated using the Promotion and ActualPrice

      <LS_KFTYPE_NEW> = 'SALEPRICE'.

      L_calc = ( L_DISCNTPCT * L_ACTPRICE ) * -1.

      <LS_SIGNEDDATA_NEW> = L_ACTPRICE - L_calc.

      APPEND <LS_NEW_REC> TO <LT_FINAL>.

    ENDLOOP.

  ENDLOOP.

** Now Add the entries to the CT_DATA.

  CT_DATA = <LT_FINAL>.

ENDMETHOD.

